

6.0 フィルター 要因分解

ver.0.1

時系列データを扱う場合、その多くは、トレンド (T)、サイクル (C)、季節変動 (S) そしてランダム項 (R) に分解できるものと仮定するとうまく行くことが多い。いまドライブ D: に datafile18.txt があるものとする。

線形トレンドの乗法モデル $X_i = S_i \times T_i \times C_i \times R_i$

T、C、S、R に分解する要素分解の代表的な行い方に、移動平均 MA 系列の線形トレンドを T と仮定する方法がある。この方法のやり方の概要は次のようになる。

- 1) S R の系列を N 期の中央移動平均で求める。ただし、N が偶数の場合、例えば、N = 4 期の場合には、

$$(X_{i-2} + X_{i-1} + X_i + X_{i+1}) \div 4 \quad \text{と} \quad (X_{i-1} + X_i + X_{i+1} + X_{i+2}) \div 4$$

の平均値を中央移動平均値とする。元のデータ X をこの中央移動平均 MA で割った値を S R とする。つまり、

$$S R = X \div M A$$

- 2) S + R からなる S R の N 期おきのそれぞれのデータの平均を求め、それを季節変動 S とする。この S は N 期ごとに一定の値となる。

- 3) MA の系列について時間に関する線形回帰を行い、その推定値をトレンド T とする。ここでは、便宜上、一直線となる線形のトレンドを仮定していることになる。

- 4) MA の系列を T で割ることによって、サイクル C を求める。すなわち、

$$M A = T \times C$$

となっているものの、C の部分を求める。

- 5) 最後に、X を T と C と S で割った残りの部分が、ランダム項 R となる。S R を S で割っても同じことである。ここで、

$$\begin{aligned} X &= M A \times S R \\ &= (T \times C) \times (S \times R) \end{aligned}$$

の R の部分を最後に求めていることになる。

プログラム

```
new; cls;
load x[24,1]=D:\datafile18.txt;
t0=1995; step=1/4; N=4;
call STCR(x,t0,step,N);

proc STCR(x,t0,step,N);
```

```

local nx,MA,i,j,SR,SS,x1,b,count,S,y,T,C,R,label;
nx=rows(x);
/* MA calculation */
MA=miss(zeros(nx,1),0);
if (N-1)%2;          /* When N is odd. */
    i=N/2+1;
    do while i<=nx-N/2;
        MA[i]=(sumc(x[i-N/2:i+N/2-1])/N+sumc(x[i-N/2+1:i+N/2])/N)/2;
        i=i+1;
    endo;
else;                /* When N is even. */
    i=(N-1)/2+1;
    do while i<=nx-(N-1)/2;
        MA[i]=sumc(x[i-(N-1)/2:i+(N-1)/2])/N;
        i=i+1;
    endo;
endif;
/* S calculation */
SR=x./MA;
SS=zeros(N,1);
count=zeros(N,1);
if (N-1)%2;
    j=1;
    do while j<=N;
        i=N/2+1;
        do while i+(j-1)<=nx-N/2;
            SS[j]=SS[j]+SR[i+(j-1)];
            count[j]=count[j]+1;
            i=i+N;
        endo;
        j=j+1;
    endo;
    SS=SS./count;
    S=miss(zeros(nx,1),0);
    j=1;
    do while j<=N;

```

```

        i=N/2+1;
        do while i+(j-1)<=nx-N/2;
            S[i+(j-1)]=SS[j];
            i=i+N;
        endo;
        j=j+1;
    endo;
else;
    j=1;
    do while j<=N;
        i=(N-1)/2+1;
        do while i+(j-1)<=nx-(N-1)/2;
            SS[j]=SS[j]+SR[i+(j-1)];
            count[j]=count[j]+1;
            i=i+N;
        endo;
        j=j+1;
    endo;
    SS=SS./count;
    S=miss(zeros(nx,1),0);
    j=1;
    do while j<=N;
        i=(N-1)/2+1;
        do while i+(j-1)<=nx-(N-1)/2;
            S[i+(j-1)]=SS[j];
            i=i+N;
        endo;
        j=j+1;
    endo;
endif;
/* T calculation */
y=delif(MA,MA.==miss(0,0));
x1=ones(rows(y),1)~seqa(1,1,rows(y));
b=inv(x1'x1)*x1'y;
T=x1*b;
if (N-1)%2;

```

```

        T=miss(zeros(N/2,1),0) | T | miss(zeros(N/2,1),0);
    else;
        T=miss(zeros((N-1)/2,1),0) | T | miss(zeros((N-1)/2,1),0);
    endif;
/* C & R calculation */
    C=MA./T;
    R=x./(S.*T.*C);
/* display the result */
    print/lz "N=" N;
    label={"x" "S" "T" "C" "R"};
    print $label;;
    print x~S~T~C~R;
/* graph */
    library pgraph;
    graphset;
    pqgwin auto;
    _plctrl={-1,0,0};
    _plegctl=1;
    _plegstr="data¥000MA¥000T";
    _pcolor={10,11,15};
    ylabel("data, MA, T");
    xy(seqa(t0,step,nx),x~MA~T);
    _plctrl={0,0,0};
    _plegctl=1;
    _plegstr="C¥000S¥000R";
    _pcolor={10,11,12};
    ylabel("C, S, R");
    xy(seqa(t0,step,nx),C~S~R);
    retp(T.*C.*R);
endp;

```

なお、入力パラメータのうち、t0 はデータの始点の年または時間、step は4半期データなら1/4 または0.25 となる。N = 4 と設定しているので、季節変動は4半期ごとに変動するものと仮定している。



画面表示

N= 4

x	S	T	C	R
15391.000
14011.000
16930.000	1.0017319	15900.921	0.98600418	1.0779643
16588.000	0.99367803	15856.199	1.0105826	1.0417833
14978.000	1.0119826	15811.477	1.0236472	0.91444591
17189.000	1.0042323	15766.755	1.0151898	1.0693671
15043.000	1.0017319	15722.033	1.0269982	0.93004624
17042.000	0.99367803	15677.311	1.0266508	1.0655664
15646.000	1.0119826	15632.589	1.0191290	0.97044340
16110.000	1.0042323	15587.867	0.99822827	1.0309671
14814.000	1.0017319	15543.145	0.97711882	0.97372099
14300.000	0.99367803	15498.423	0.96415452	0.96306630
15406.000	1.0119826	15453.702	0.96022302	1.0259170
14393.000	1.0042323	15408.980	0.97595691	0.95304328
15700.000	1.0017319	15364.258	0.98435443	1.0362989
15010.000	0.99367803	15319.536	0.98700773	0.99900773
15379.000	1.0119826	15274.814	0.98065027	1.0145302
14393.000	1.0042323	15230.092	0.96646661	0.97370584
14570.000	1.0017319	15185.370	0.97540432	0.98196949
14061.000	0.99367803	15140.648	1.0043246	0.93057625
17068.000	1.0119826	15095.926	1.0396265	1.0746635
15858.000	1.0042323	15051.204	1.0783689	0.97291667
17009.000
15915.000

上の画面結果のように、元のデータ x から線形のタイムトレンドが求められ、その他の要素のサイクル C 、季節変動 S 、それにランダム項 R は 1 のまわりの値になっていることがわかる。また、 S は $N = 4$ に設定されているので、4 期ごとに同じパターンの値になる。なお、次のように、 C と S と R のプロットをしてやると、この要素分解によって、サイクル C がどのように循環しているかが一目両全となる。

グラフ表示



線形トレンドの乗法モデル(S が加法的なケース) $X_i = S_i + (T_i \times C_i \times R_i)$

季節変動 S の項が、加法的に加わっていると仮定すると、プログラムの当該部分だけを変更するだけで、同じことを行なうことができる。

プログラム (変更部分の前後のみ)

/* C & R calculation */

$C = MA / T$;



$R = (x - S) / (T * C)$; /* Assume $x = S + (T * C * R)$. */

画面表示

N= 4

x	S	T	C	R
15391.000
14011.000
16930.000	1.0017319	15900.921	0.98600418	1.0797674
16588.000	0.99367803	15856.199	1.0105826	1.0351352
14978.000	1.0119826	15811.477	1.0236472	0.92534081
17189.000	1.0042323	15766.755	1.0151898	1.0738303
15043.000	1.0017319	15722.033	1.0269982	0.93159498
17042.000	0.99367803	15677.311	1.0266508	1.0587682
15646.000	1.0119826	15632.589	1.0191290	0.98200830
16110.000	1.0042323	15587.867	0.99822827	1.0352659
14814.000	1.0017319	15543.145	0.97711882	0.97534145
14300.000	0.99367803	15498.423	0.96415452	0.95691133
15406.000	1.0119826	15453.702	0.96022302	1.0381419
14393.000	1.0042323	15408.980	0.97595691	0.95701006
15700.000	1.0017319	15364.258	0.98435443	1.0380275
15010.000	0.99367803	15319.536	0.98700773	0.99262632
15379.000	1.0119826	15274.814	0.98065027	1.0266194
14393.000	1.0042323	15230.092	0.96646661	0.97775862
14570.000	1.0017319	15185.370	0.97540432	0.98360257
14061.000	0.99367803	15140.648	1.0043246	0.92462783
17068.000	1.0119826	15095.926	1.0396265	1.0874762
15858.000	1.0042323	15051.204	1.0783689	0.97697246
17009.000
15915.000

グラフ表示



なお、結果は省略をするけれども、一番最初の乗法モデルを仮定したときに、S Rの系列から季節変動Sを求める際に、N期ごとのそれぞれの最大値と最小値を除くことにより、季節変動Sを調整してやったプログラムは以下ようになる。

プログラム

```
new; cls;
load x[24,1]=D:\datafile18.txt;
t0=1995; step=1/4; N=4;
call STCRa(x,t0,step,N);
```



```

proc STCRa(x,t0,step,N);
    local nx,MA,i,j,SR,SS,x1,b,Sx,S,y,T,C,R,label;
    nx=rows(x);
/* MA calculation */
    MA=miss(zeros(nx,1),0);
    if (N-1)%2;
        i=N/2+1;
        do while i<=nx-N/2;
            MA[i]=(sumc(x[i-N/2:i+N/2-1])/N+sumc(x[i-N/2+1:i+N/2])/N)/2;
            i=i+1;
        endo;
    else;
        i=(N-1)/2+1;
        do while i<=nx-(N-1)/2;
            MA[i]=sumc(x[i-(N-1)/2:i+(N-1)/2])/N;
            i=i+1;
        endo;
    endif;
/* S calculation */
    SR=x./MA;
    SS=zeros(N,1);
    if (N-1)%2;
        j=1;
        do while j<=N;
            Sx=miss(0,0);
            i=N/2+1;
            do while i+(j-1)<=nx-N/2;
                Sx=Sx | SR[i+(j-1)];
                i=i+N;
            endo;
            Sx=Sx[2:rows(Sx)];
            Sx=delif(Sx,Sx.==maxc(Sx));
            Sx=delif(Sx,Sx.==minc(Sx));
            SS[j]=meanc(Sx);
            j=j+1;
        endo;
    else;
        j=1;
        do while j<=N;
            Sx=miss(0,0);
            i=(N-1)/2+1;
            do while i+(j-1)<=nx-(N-1)/2;
                Sx=Sx | SR[i+(j-1)];
                i=i+N;
            endo;
            Sx=Sx[2:rows(Sx)];
            Sx=delif(Sx,Sx.==maxc(Sx));
            Sx=delif(Sx,Sx.==minc(Sx));
            SS[j]=meanc(Sx);
            j=j+1;
        endo;
    endif;
endproc;

```

```

S=miss(zeros(nx,1),0);
j=1;
do while j<=N;
    i=N/2+1;
    do while i+(j-1)<=nx-N/2;
        S[i+(j-1)]=SS[j];
        i=i+N;
    endo;
    j=j+1;
endo;
else;
j=1;
do while j<=N;
    Sx=miss(0,0);
    i=(N-1)/2+1;
    do while i+(j-1)<=nx-(N-1)/2;
        Sx=Sx | SR[i+(j-1)];
        i=i+N;
    endo;
    Sx=Sx[2:rows(Sx)];
    Sx=delif(Sx,Sx.==maxc(Sx));
    Sx=delif(Sx,Sx.==minc(Sx));
    SS[j]=meanc(Sx);
    j=j+1;
endo;
S=miss(zeros(nx,1),0);
j=1;
do while j<=N;
    i=(N-1)/2+1;
    do while i+(j-1)<=nx-(N-1)/2;
        S[i+(j-1)]=SS[j];
        i=i+N;
    endo;
    j=j+1;
endo;
endif;

```

```

/* T calculation */
y=delif(MA,MA.==miss(0,0));
x1=ones(rows(y),1)~seqa(1,1,rows(y));
b=inv(x1'x1)*x1'y;
T=x1*b;
if (N-1)%2;
    T=miss(zeros(N/2,1),0) | T | miss(zeros(N/2,1),0);
else;
    T=miss(zeros((N-1)/2,1),0) | T | miss(zeros((N-1)/2,1),0);
endif;
/* C & R calculation */
C=MA./T;
R=x./(S.*T.*C);
/* display the result */
print/lz "N=" N;
label={"x" "S" "T" "C" "R"};
print $label;;
print x~S~T~C~R;
/* graph */
library pgraph;
graphset;
pqgwin auto;
    _plctrl={-1,0,0};
    _plegctl=1;
    _plegstr="data¥000MA¥000T";
    _pcolor={10,11,15};
    ylabel("data, MA, T");
    xy(seqa(t0,step,nx),x~MA~T);
    _plctrl={0,0,0};
    _plegctl=1;
    _plegstr="C¥000S¥000R";
    _pcolor={10,11,12};
    ylabel("C, S, R");
    xy(seqa(t0,step,nx),C~S~R);
    retp(T.*C.*R);
endp;

```

なお、季節変動を加法的に組み入れたケースについても、同様の箇所を変更することで簡単に求められるだろう。

高次の多項トレンドの乗法モデル $X_i = S_i \times T_i \times C_i \times R_i$

次に、線形トレンドを仮定しない変更を加えたものを見ていこう。ここでは、線形回帰ではなくて、多項式を用いた回帰でもって、高次のトレンドをもったT成分を考えることでプログラムを変更してやろう。

プログラム

```
new; cls;
load x[24,1]=D:\datafile18.txt;
t0=1995; step=1/4; N=4; porder=7;
call STCRp(x,t0,step,N,porder);

proc STCRp(x,t0,step,N,porder);
    local nx,MA,i,j,SR,SS,Sx,S,y,T,C,R,label;
    nx=rows(x);
/* MA calculation */
    MA=miss(zeros(nx,1),0);
    if (N-1)%2;
        i=N/2+1;
        do while i<=nx-N/2;
            MA[i]=(sumc(x[i-N/2:i+N/2-1])/N+sumc(x[i-N/2+1:i+N/2])/N)/2;
            i=i+1;
        endo;
    else;
        i=(N-1)/2+1;
        do while i<=nx-(N-1)/2;
            MA[i]=sumc(x[i-(N-1)/2:i+(N-1)/2])/N;
            i=i+1;
        endo;
    endif;
/* S calculation */
    SR=x./MA;
    SS=zeros(N,1);
    if (N-1)%2;
```

```

j=1;
do while j<=N;
    Sx=miss(0,0);
    i=N/2+1;
    do while i+(j-1)<=nx-N/2;
        Sx=Sx | SR[i+(j-1)];
        i=i+N;
    endo;
    Sx=Sx[2:rows(Sx)];
    Sx=delif(Sx,Sx.==maxc(Sx));
    Sx=delif(Sx,Sx.==minc(Sx));
    SS[j]=meanc(Sx);
    j=j+1;
endo;
S=miss(zeros(nx,1),0);
j=1;
do while j<=N;
    i=N/2+1;
    do while i+(j-1)<=nx-N/2;
        S[i+(j-1)]=SS[j];
        i=i+N;
    endo;
    j=j+1;
endo;
else;
j=1;
do while j<=N;
    Sx=miss(0,0);
    i=(N-1)/2+1;
    do while i+(j-1)<=nx-(N-1)/2;
        Sx=Sx | SR[i+(j-1)];
        i=i+N;
    endo;
    Sx=Sx[2:rows(Sx)];
    Sx=delif(Sx,Sx.==maxc(Sx));
    Sx=delif(Sx,Sx.==minc(Sx));

```

```

        SS[j]=meanc(Sx);
        j=j+1;
    endo;
    S=miss(zeros(nx,1),0);
    j=1;
    do while j<=N;
        i=(N-1)/2+1;
        do while i+(j-1)<=nx-(N-1)/2;
            S[i+(j-1)]=SS[j];
            i=i+N;
        endo;
        j=j+1;
    endo;
endif;
/* T calculation */
y=delif(MA,MA.==miss(0,0));
T=trendp(y,sega(1,1,rows(y)),porder);
if (N-1)%2;
    T=miss(zeros(N/2,1),0) | T | miss(zeros(N/2,1),0);
else;
    T=miss(zeros((N-1)/2,1),0) | T | miss(zeros((N-1)/2,1),0);
endif;
/* C & R calculation */
C=MA./T;
R=x./(S.*T.*C);
/* display the result */
print/lz "N=" N;
label={"x" "S" "T" "C" "R"};
print $label;;
print x~S~T~C~R;
/* graph */
library pgraph;
graphset;
pqgwin auto;
    _plctrl={-1,0,0};
    _plegctl=1;

```

```

    _plegstr="data¥000MA¥000T";
    _pcolor={10,11,15};
        ylabel("data, MA, T");
    xy(seqa(t0,step,nx),x~MA~T);
    _plctrl={0,0,0};
    _plegctl=1;
    _plegstr="C¥000S¥000R";
    _pcolor={10,11,12};
        ylabel("C, S, R");
    xy(seqa(t0,step,nx),C~S~R);
    retp(T.*C.*R);
endp;

```



```

proc trendp(y,x,porder);
    local n,i,xx,beta,yhat;
    n=rows(x);
    xx=ones(n,1);
    i=1;
    do while i<=porder;
        xx=xx~(x^i);
        i=i+1;
    endo;
    beta=inv(xx'xx)*xx'y;
    yhat=xx*beta;
    retp(yhat);
endp;

```

上では、N期とびのS Rの系列からそれぞれ最大値と最小値を取り除く形で、その平均を見ることで、季節変動Sを求めているプログラムと同じである。ただし、MAの系列の線形のトレンドをトレンドTにするのではなくて、別途 trendp という procedure を作ってやって、porder の次数の高次多項トレンドを計算している。その部分のプログラムだけが違っている。なお、データの長さにもよるが、porder の次数の大きさをあまり大きくすることは計算上できない。Trendp の内部で高次の項が指数的に爆発をしてオーバーフローしてしまうからである。

グラフ表示



移動平均MAに対して、線形のトレンドではなくて、多項式を念頭に置いたトレンドを計算していることがわかるだろう。もちろん、トレンドT自体がサイクルを描いており、このデータは短い期間を対象としたものであり、極端な計算例として取り上げているだけである。あまり現実的ではない。

線形トレンドの加法モデル $X_i = S_i + T_i + C_i + R_i$

最後の polynomial トレンドを用いたバージョン(もちろん porder = 1 と設定すれば線形トレンドを仮定して計算することになる) の加法モデルも計算してみよう。

$$S R = X - M A$$

$$M A = T + C$$

というふうに割り算のところを引き算に変更するだけでできる。なお、ここでは、

$$\begin{aligned} X &= M A + S R \\ &= (T + C) + (S + R) \end{aligned}$$

だから、最終的にランダム項 R を求めるには元データから (S + T + C) を引いてやればよいことになる。

以下では、porder=3 の時の 3 次トレンドを仮定した加法モデルの例を示しておこう。繰り返しになるが、porder=1 と設定すれば、線形トレンドを仮定した結果も表示される。

プログラム

```
new; cls;
```

```
load x[24,1]=D:\datafile18.txt;
```

```
t0=1995; step=1/4; N=4; porder=3;
```

```
call Add_STCRp(x,t0,step,N,porder);
```

```
proc Add_STCRp(x,t0,step,N,porder);
```

```
local nx,MA,i,j,SR,SS,Sx,S,y,T,C,R,label;
```

```
nx=rows(x);
```

```
/* MA calculation */
```

```
MA=miss(zeros(nx,1),0);
```

```
if (N-1)%2;
```

```
    i=N/2+1;
```

```
    do while i<=nx-N/2;
```

```
        MA[i]=(sumc(x[i-N/2:i+N/2-1])/N+sumc(x[i-N/2+1:i+N/2])/N)/2;
```

```
        i=i+1;
```

```
    endo;
```

```
else;
```

```
    i=(N-1)/2+1;
```

```
    do while i<=nx-(N-1)/2;
```

```
        MA[i]=sumc(x[i-(N-1)/2:i+(N-1)/2])/N;
```

```
        i=i+1;
```

```
    endo;
```

```

    endif;
/* S calculation */
    SR=x-MA;
    SS=zeros(N,1);
    if (N-1)%2;
        j=1;
        do while j<=N;
            Sx=miss(0,0);
            i=N/2+1;
            do while i+(j-1)<=nx-N/2;
                Sx=Sx | SR[i+(j-1)];
                i=i+N;
            endo;
            Sx=Sx[2:rows(Sx)];
            Sx=delif(Sx,Sx.==maxc(Sx));
            Sx=delif(Sx,Sx.==minc(Sx));
            SS[j]=meanc(Sx);
            j=j+1;
        endo;
        S=miss(zeros(nx,1),0);
        j=1;
        do while j<=N;
            i=N/2+1;
            do while i+(j-1)<=nx-N/2;
                S[i+(j-1)]=SS[j];
                i=i+N;
            endo;
            j=j+1;
        endo;
    else;
        j=1;
        do while j<=N;
            Sx=miss(0,0);
            i=(N-1)/2+1;
            do while i+(j-1)<=nx-(N-1)/2;
                Sx=Sx | SR[i+(j-1)];

```

```

/* Additive Model */

```

```

        i=i+N;
    endo;
    Sx=Sx[2:rows(Sx)];
    Sx=delif(Sx,Sx.==maxc(Sx));
    Sx=delif(Sx,Sx.==minc(Sx));
    SS[j]=meanc(Sx);
    j=j+1;
end;
S=miss(zeros(nx,1),0);
j=1;
do while j<=N;
    i=(N-1)/2+1;
    do while i+(j-1)<=nx-(N-1)/2;
        S[i+(j-1)]=SS[j];
        i=i+N;
    end;
    j=j+1;
end;
endif;
/* T calculation */
y=delif(MA,MA.==miss(0,0));
T=trendp(y,seqa(1,1,rows(y)),porder);
if (N-1)%2;
    T=miss(zeros(N/2,1),0) | T | miss(zeros(N/2,1),0);
else;
    T=miss(zeros((N-1)/2,1),0) | T | miss(zeros((N-1)/2,1),0);
endif;
/* C & R calculation */
C=MA-T;
R=x-(S+T+C);
/* display the result */
print/lz "N=" N;
label={"x" "S" "T" "C" "R"};
print $label;;
print x~S~T~C~R;
/* graph */
/* Additive Model */
/* Additive Model */

```

```

library pgraph;
graphset;
pqgwin auto;
  _plctrl={-1,0,0};
  _plegctl=1;
  _plegstr="data¥000MA¥000T";
  _pcolor={10,11,15};
  ylabel("data, MA, T");
  xy(seqa(t0,step,nx),x~MA~T);
  _plctrl={0,0,0};
  _plegctl=1;
  _plegstr="C¥000S¥000R";
  _pcolor={10,11,12};
  ylabel("C, S, R");
  xy(seqa(t0,step,nx),C~S~R);
  retp(T+C+R);
endp;



```

```

proc trendp(y,x,porder);
  local n,i,xx,beta,yhat;
  n=rows(x);
  xx=ones(n,1);
  i=1;
  do while i<=porder;
    xx=xx~(x^i);
    i=i+1;
  endo;
  beta=inv(xx'xx)*xx'y;
  yhat=xx*beta;
  retp(yhat);
endp;

```

画面表示



上のプロットのように、加法モデルの場合には、CとSとRの軌跡が、1のまわりではなくて、今度は0のまわりに描かれている。

なお、この章で取り上げた procedure はすべて call で止めているが、これを変数で受けるか、直接 print 文で画面表示させれば、乗法モデルのケースでは $T+C+R$ が、加法モデルでは $T+C+R$ が求められる（すでに procedure のアウトプットとして retp の中に設定されている）。これらが、季節変動調整済みデータと呼ばれるものである。